
Bascom Code. 

$regfile = "attiny13.dat" 
$crystal = 600000                                           ' Set 4.8 MHz Clock, Divider = 8 
$hwstack = 16 
$swstack = 16 
$framesize = 16 

Config Portb.0 = Output                                     'grüne LED 
Config Portb.1 = Input                                      'Taster an Masse 
Config Portb.2 = Output                                     'rote LED 
Config Portb.3 = Output                                     'Power Ausgang 
Config Portb.4 = Input                                      'ADC Akkuspannung 

Portb.1 = 1                                                 'Pullupwiderstand 

Const Adc_ergebnis = 0.0048828125                           '= 5,0 / 1024,0 

Enable Interrupts 

Dim Messergebnis As Word 
Dim Volt As Single 
Dim A As Byte 
Dim B As Byte 

Config Adc = Single , Prescaler = Auto , Reference = Avcc 
Start Adc 

Gn Alias Portb.0 
Rt Alias Portb.2 
Taster Alias Pinb.1 
Pwr Alias Portb.3 

 
Switchon:                                                   'Schalter eingeschalten 

   Reset Rt 
   Set Gn 
   Set Pwr 
   Wait 1                                                   'benötigte Zeit um Taster loszulassen 
   If Taster = 0 Then 
      Goto Sicherung 
   End If 

Goto Switchon: 

Sicherung: 

   B = 0                                                    'setzt Zählwert auf 0 
   Reset Gn 
   Set Pwr 
   Set Rt 
   Waitms 50 
   Reset Rt 
   Waitms 50 

   Incr A 

   If Taster = 0 And A > 14 Then                            '14 = Zeitfaktor 
      A = 0 
      Wait 2                                                'benötigte Zeit um Taster loszulassen 
      Goto Switchoff 
   End If 

   If Taster = 1 Then 
      A = 0 
      Goto Switchon 
   End If 



Goto Sicherung: 

Switchoff:                                                  'Schalter ausgeschalten, Akkuüberwachung aktiv 

   Config Adc = Single , Prescaler = Auto , Reference = Avcc 
   Start Adc 

   Messergebnis = Getadc(2) 
   Volt = Messergebnis * Adc_ergebnis 

   Reset Gn 
   Reset Pwr 
   Reset Rt 

   Select Case Volt 

      Case Is < 3.5                                         'Akku ist leer 
      Set Rt 
      Waitms 50 
      Reset Rt 

      Case Is < 3.8                                         'Akku ist halbvoll 
      Set Gn 
      Waitms 50 
      Reset Gn 
      Waitms 150 
      Set Rt 
      Waitms 50 
      Reset Rt 

      Case Is < 5.0                                         'Akku ist voll 
      Set Gn 
      Waitms 50 
      Reset Gn 

   End Select 

   If Taster = 0 Then 
      Goto Switchon 
   End If 

   Incr B 

   If B > 30 Then                                           'nach 30 Zyklen ( ca 1 Min.) springt Programm in 
      Goto Aus                                                'den Schlafmodi 
   End If 

   Wait 2                                                   'Akkuanzeige wird alle 2 Sek. angezeigt 

Goto Switchoff: 

Aus: 
   Stop Adc                                                 'Programm schläft 4 Sek., wacht auf, durchläuft 
   Stop Ac                                                  'die Akkuüberwachung und schläft wieder 4 Sek. 
   Wdtcr = &B11110000                                       'Stromverbrauch ~ 15 µA 
   Reset Watchdog 
   Powerdown 
   Wdtcr = &B11010011 
Goto Switchoff 

End 

 


